Circuit Simulation as an Aid in Teaching the Principles of Power Electronics

Daniel W. Hart, *Senior Member, IEEE*

Abstract—**This paper describes how computer simulation can assist power electronics education at the undergraduate level. Using PSpice as an instructional tool for an introductory-level power electronics course has increased students' abilities to comprehend the behavior of several power electronics circuits. The graphics post-processor Probe, which accompanies PSpice, is used to display voltage and current waveforms. Probe's capability of doing numerical computations reduces the effort needed to determine power-related quantities in nonlinear circuits. Fourier analysis capabilities of PSpice enable harmonics to be investigated. Examples of PSpice simulations are included.**

I. Introduction

An overall objective of a power electronics circuit is to match the voltage and current requirements of the load to the source. Power electronics circuits function by using semiconductor devices as switches, thereby modifying an input voltage. Advances in semiconductor switching capability, combined with the desire to improve the efficiency and performance of electrical devices, are making power electronics a fast-growing area in electrical engineering.

Circuit simulation has become an integral part of several courses, and the adaption of PSpice to the power electronics course has been a natural evolvement for the students and for the author. At the author's institution, students are familiar with PSpice by the time they take the power electronics course, which is an elective for undergraduate seniors. PSpice is especially useful for students because it is designed for the personal computer, now a standard item for electronical engineering students. An evaluation version of PSpice is adequate for most student projects and is available at no cost through MicroSim Corporation.[1] All simulations described in this paper can be run using the evaluation version.

An increase in the understanding of several power electronics circuits is possible through simulation using PSpice. The focus of this discussion is on using PSpice for the instructional benefit of investigating the basic behavior of power electronics circuits using idealized components, realizing that the results are first-order approximations, much the same as the analytical work done in the first discussion of the subject in a textbook.

Using Probe, the graphics post-processor program that accompanies PSpice, the waveform of any current and voltage in the circuit can be shown graphically. Moreover, Probe is capable of performing mathematical computations involving currents and/or voltages, including numerical determination of RMS values and average power [1].

II. The Need for Computer Simulation of Power-Electronics Circuits

Observing the voltage and current waveforms in a circuit can provide an insight that is not always possible with pencil-and-paper analysis. Simulation may be used in place of a laboratory experiment, especially during the student's initial study of power electronics circuits, or simulation can be used as a design tool in preparation for laboratory work [2].

Observing voltage and current waveforms from a computer simulation accomplishes some of the same objectives as those of a laboratory experience. In a computer simulation, all of the circuit's voltages and currents can be investigated, usually much more efficiently than in a hardware laboratory. Variations in circuit performance produced by a change in components or operating parameters can be accomplished more easily with computer simulation than in a laboratory. In a laboratory exercise, problems may arise with components, equipment, and wiring, which detract from the primary objective of understanding the operation of a circuit, and, in some instances, a laboratory experiment is not practical. High-power equipment may not be available for student use, and suitable instrumentation must be employed for power measurements and spectrum analysis.

III. Circuit Modeling

The choice of a circuit model depends upon the objectives of the simulation. If the goal is to predict the behavior of a circuit before it is built, components should be modeled in detail. However, if the goal is to study a circuit to gain an understanding of the principles of operation, component models should be kept as elementary as possible. In some cases, devices can be modeled as ideal components. For example, a voltage-controlled switch can be used in place of a transistor to simulate a switching operation. In other cases, it is most convenient to use default models for devices such as the diode, transistor, and SCR. Drive circuits for transistors and SCR's are idealized. The results of simulations using idealized or default components are first approximations of circuit behavior, but are useful for instructional purposes. Because it is highly desirable that the student be able to run the simulations on his or her own computer for homework, the limitations of the evaluation version of PSpice should also be considered when choosing a circuit model.

Manuscript received June 1992.
The author is with the Department of Electrical and Computer Engineering, Valparaiso University, Valparaiso, IN 46383.
IEEE Log Number 9202868.

[1] MicroSim Corporation, 20 Fairbanks, Irvine, CA 92718.

Switching functions are essential in power electronics circuits. Switches are realized with BJT's, MOSFET's, SCR's, GTO's, IGBT's, or other devices. Over recent years, the preferred switching device for a particular circuit has changed with advancements in solid-state technology. What has remained fundamental to power electronics is the theory of overall circuit performance when switching occurs. Therefore, switch models used in introductory simulations should be kept as simple as possible.

Electronic switches are simplified for initial circuit simulation by using voltage-controlled switches and diodes. The voltage-controlled switch is a standard component in PSpice, and the diode used most often is the default diode model. In some cases, students prefer to use an "ideal" diode to obtain results which they can compare with the theoretical. This is accomplished with a diode model which has the diode equation parameter $N = (10)^{-4}$, rather than the default value of 1, producing a forward voltage drop in the microvolt range. Although not elaborate and missing some notable characteristics, these switch models provide a means of simulating the basic switching functions for the purpose of analyzing overall circuit performance. The advantage of this approach is that students will gain an overall understanding of the intended operation of a circuit before going on to more realistic device modeling.

The evaluation version of PSpice (version 5.0) now includes an SCR model in the device library which can be used in a small circuit such as the half-wave rectifier. However, the evaluation version, which is limited in the allowable number of nodes and devices, will not accept more than one SCR in a circuit. A simplified model of the SCR is a voltage-controlled switch in series with a diode. With this SCR model, the diode allows current in the proper direction and the switch is used as a control. A drawback is that the switch must be kept closed at least for the conduction duration. Improved SCR models include a current latching provision that could be implemented by a current-sensing resistor or voltage source [3] and [4], but these enhancements add to the total number of nodes and devices in the circuit.

Transformers can be modeled using inductances which have ideal magnetic coupling. Nonideal transformer coupling may be added to show the need for snubber circuits to protect the switches.

IV. Power Computations

Power computations are an important part of power electronics, but power computations for nonlinear circuits are sometimes confusing to students. To calculate analytically the average power supplied to a load for some circuits, it may be necessary to develop an expression for load current, compute the RMS value from a difficult integration, and finally determine I^2R [5]–[7]. Before using PSpice in the power electronics class, much valuable time was spent on analytic and numerical techniques for determining power quantities. With the adoption of PSpice, RMS, and power calculations are often done with the software.

In addition to computing voltages and currents, the numerical capabilities of the graphics post-processor program Probe can be used to determine power relationships based on voltages and currents in a circuit [1]. Students can determine power quantities in somewhat complicated circuits by relying on basic power definitions without becoming overwhelmed with the mathematics. Simulation thus enables students to determine power quantities which would not be practically obtainable by analytic methods. An example of this technique is given below.



Fig. 1. A half-wave rectifier.

V. Examples

A. A Controlled Half-Wave Rectifier

The first topic in the power electronics course is the single-phase half-wave rectifier. Single-phase rectifiers are introduced in the first electronics course, but the loads there are restricted to resistors or a resistance with a capacitor filter. In power electronics, the SCR may be used instead of the diode, and loads are more general. An example is shown in Fig. 1.

The model for the SCR is a voltage-controlled switch with a series diode. The switch energizes the load at the specified time, and the diode allows the current to flow in only the forward direction, thus simulating the SCR. The switch control must keep the switch closed for at least as long as the SCR is forward biased. The PSpice SCR model could be used with a simple gate control circuit.

The PSpice input file for the simulation of this circuit is shown at the top of the next page.

The resulting voltage and magnified current are shown in Fig. 2, which is the output from Probe. The average power absorbed by the resistor and DC source are computed by entering the expressions AVG($V(3,4)*I(R)$) and AVG ($V(5)*I(VC)$), respectively. Note that Probe gives a "running" value of any computation, so the appropriate number is obtained at the end of one period of the waveform. The power absorbed by the resistor is 23.5 W, and that by the DC source is 46.5 W. Similarly, the average and RMS values of the current are determined by entering the expressions AVG ($I(R)$) and RMS ($I(R)$) when in Probe.

Without the use of PSpice, an analytic computation of RMS current for this circuit would be accomplished by evaluating the expression

$$I_{\mathrm{RMS}} = \sqrt{\frac{1}{2\pi}\int_0^{2\pi} i^2(\omega t)\, d(\omega t)}$$

$$= \sqrt{\frac{1}{2\pi}\int_\alpha^\beta \left[4.35\sin(\omega t - 1.31) - 5 + 8.1e^{-\omega t/3.77}\right]^2 d(\omega t)}$$

where $\alpha = 60^\circ = 1.05$ rad., and β is found from the numerical

```
HALF-WAVE RECTIFIER EXAMPLE
*Simplified SCR model
.PARAM ALPHA = 60        ; delay angle of SCR
.PARAM DELAY = {ALPHA/360/60} PW = {1/60-DELAY}
VS 1 0 SIN(0 170 60)
DSCR 1 2 DMOD            ; diode part of SCR
SSCR 2 3 6 0 SMOD        ; switch part of SCR
R 3 4 10
L 4 5 100MH IC = 0
VC 5 0 DC 50
VCONTROL 6 0 PULSE(-1 1 {DELAY} 1 US 1US {PW} {1/60})
.MODEL DMOD                                      ; default diode
.MODEL SMOD VSWITCH (RON = .001 VON = .001) ; voltage-controlled switch
.TRAN .1MS 16.67MS UIC                           ; transient analysis
.PROBE
.END
```



Fig. 2. a) Source voltage and current. b) Average power.

solution of

$$4.35\sin(\beta - 1.31) + 0.26e^{(1.05-\beta)/3.77} = 0.$$



Fig. 3. A controlled three-phase rectifier.

The time-saving benefit of a PSpice simulation to determine power becomes obvious from the above example.

B. A Controlled Three-Phase Rectifier

In the controlled three-phase rectifier shown in Fig. 3, the switches which simulate the SCR's must be properly sequenced. Again, the SCR model is the voltage-controlled switch with a series diode. The circuit input file for a three-phase rectifier is shown at the top of the next page.

The output voltage and one SCR voltage are shown in Fig. 4.

With this simulation, the student can observe all voltages and currents in the circuit, including each of the SCR's. The Fourier series of the load current and source line current will be in the output file if the .FOUR command is included. Loads can be changed, and source inductance can be added to investigate the commutation between switches. With some modification, a second six-pulse bridge can be added to the above file to simulate a twelve-pulse converter using the evaluation version.

C. A DC–DC Converter

It is useful for the student to investigate the basic operation of DC–DC converter circuits using PSpice circuit models that are based on ideal components. A circuit for which simulation increases student understanding is the forward DC–DC converter shown in Fig. 5. The periodic opening and closing of the switch enables energy to be transferred from the source to the load. What is generally confusing to the students is the role of each of the transformer windings. A simulation allows all of the voltage and current waveforms to be observed and shows how the third winding returns the

```
THREE-PHASE BRIDGE RECTIFIER...6 PULSE
*FOR USE WITH THE EVALUATION VERSION 5.0
.PARAM ALPHA = 30
.PARAM VM = 170 RL = 1 F = 60 DELAY = {1/(6*F)} PW = (1/(2*F)) PER = {1/F}
VAN 1 0 SIN(0 {VM} {F} 0 0 30)
VBN 2 0 SIN(0 {VM} {F} 0 0 -90
VCN 3 0 SIN(0 {VM} {F} 0 0 -210)
S1 1 8 18 0 SMOD
D1 8 4 DMOD
S4 5 9 19 0 SMOD
D4 9 1 DMOD
S3 2 10 20 0 SMOD
D3 10 4 DMOD
S6 5 11 21 0 SMOD
D6 11 2 DMOD
S5 3 12 22 0 SMOD
D5 12 4 DMOD
S2 5 13 23 0 SMOD
D2 13 3 DMOD
R 4 5 (RL); load
V1 18 0 PULSE(-10 10 {ALPHA*PER/360} 1US 1US {PW} {PER})
V4 19 0 PULSE(-10 10 {ALPHA*PER/360+3*DELAY}  1US 1US {PW} {PER})
V3 20 0 PULSE(-10 10 {ALPHA*PER/360+2*DELAY}  1US 1US {PW} {PER})
V6 21 0 PULSE(-10 10 {ALPHA*PER/360+5*DELAY}  1US 1US {PW} {PER})
V5 22 0 PULSE(-10 10 {ALPHA*PER/360+4*DELAY}  1US 1US {PW} {PER})
V2 23 0 PULSE(-10 10 {ALPHA*PER/360+DELAY}    1US 1US {PW} {PER})
.MODEL SMOD VSWITCH(RON = 0.01)
.MODEL DMOD D
.TRAN .1MS 33.33MS UIC
.FOUR 60 I(VAN) I(R)
.PROBE
.END
```



Fig. 4. Voltages for the Circuit of Fig. 3.

stored energy in the transformer magnetizing inductance to the source.



Fig. 5. Forward DC–DC converter.

The current in each of the transformer windings and the source instantaneous power are shown in Fig. 6.

The circuit used for the PSpice simulation sets the magnetizing inductance of the transformer but neglects leakage inductance and losses. The switch is a voltage-controlled switch. The PSpice input file is shown at the top of the next page.

D. An Inverter

PSpice simulation is also useful when demonstrating circuits designed for conversion of DC–AC. The full-bridge circuit of Fig. 7 is used as an inverter in the following example.

The feedback diodes, which would also be present in a real implementation, are necessary for the PSpice simulation to allow for coordination of the switches and continuity of current for inductive loads. With this switch model, switch current will

```
FORWARD SWITCHING CONVERTER
.PARAM INPUT = 30 DUTY = 0.4 RLOAD = 5 L = 1mH C = 1000UF
.PARAM FREQ = 5K                  ; Switching frequency
.PARAM LM = 2mH                   ; Magnetizing inductance (primary)
.PARAM RATIO 12 = 1               ; N1/N2 of transformer
.PARAM RATIO 13 = 1               ; N1/N3 of transformer
.TRAN 1US 400US UIC
VS 1 0 DC {INPUT}
L1 1 2 {LM} IC = 0                ; Transformer primary (N1)
L2 4 0 {LM/(RATIO12*RATIO12)}     ; Transformer secondary (N2)
L3 3 1 {LM/(RATIO13*RATIO13)}     ; Transformer tertiary (N3)
K L1 L2 L3 .9999999               ; Coupling between windings
S 2 0 20 0 SMOD                   ; Switch
D1 4 5 DMOD
D2 0 5 DMOD
D3 0 3 DMOD
LX 5 6 {L} IC = 1.5
C 6 0 {C} IC = 11.2
R 6 0 {RLOAD}
*SWITCH CONTROL:
VPULSE 20 0 PULSE (1 -1 0 .5US {DUTY/FREQ} {1/FREQ})
.MODEL SMOD VSWITCH (RON = .01 VON = -.1 VOFF = .1)
.MODEL DMOD D
.OPTIONS NOPAGE ITL5 = 0
.PROBE
.END
```



Fig. 6. Forward DC–DC converter output.

be bidirectional, unlike real switches. A series diode can solve that problem if switch currents are to be investigated, or BJT's could be used if drive circuits are included.

An application of the full-bridge inverter is pulse-width modulation (PWM) of the output voltage. One PWM scheme controls the switches by comparing a high-frequency triangular waveform with a reference sinusoid [6]. A PSpice input file that accomplishes this is shown at the top of the next page.



Fig. 7. Full-bridge inverter.

While it is useful to observe the output voltage and current waveforms, the real advantage of the simulation comes from observing the spectra of the output voltage and current. The Probe output is shown in Fig. 8, along with the Fourier series representation. The inductive load provides filtering for the harmonic frequencies, thus showing that the output current is quite sinusoidal. Students gain additional insight by first displaying the voltage output and then observing the current trace as it is being drawn.

With this PSpice input file, students can investigate other switching frequencies and amplitude modulation ratios. Other schemes, such as unipolar PWM, can be examined by modifying the switching scheme. This is a good exercise for the students to determine if they understand PWM control. Other harmonic and amplitude control switching schemes are possible with this circuit.

```
FULL-BRIDGE INVERTER- -BIPOLAR PWM
*Input parameters:
.PARAM Fo = 60                ; fundamental frequency
.PARAM Mf = 21                ; carrier, multiple of Fo
.PARAM Ma = .9                ; amplitude ratio
.PARAM Fc = {Mf*Fo}           ; carrier frequency
Vsource 1 0 DC 100
*SWITCHES
S1 1 2 40 30 SWITCH
S2 3 0 40 30 SWITCH
S3 1 3 30 40 SWITCH
S4 2 0 30 40 SWITCH
*FEEDBACK DIODES:
D1 2 1 DMOD
D2 0 3 DMOD
D3 3 1 DMOD
D4 0 2 DMOD
*LOAD:
R 2 4 1
L 4 3 2 .65MH IC = 0
*Triangle carrier
Vtri 30 0 PULSE (1 -1 0 {1/(2*Fc)} {1/(2*Fc)} 1ns {1/Fc})
*Control-include phase delay to synchronize carrier and control
Vcont 40 0 SIN(0{Ma} {Fo} 0 0 {-90/Mf})
*MODELS:
.MODEL SWITCH VSWITCH (RON = 0.001 VON = .005 VOFF = -.005)
.MODEL DMOD D
.PROBE
.TRAN 0.5MS 33.33MS 16.67MS UIC
.OPTIONS NO PAGE ITL5 = 0
.END
```



(a)



(b)

Fig. 8. Inverter output for an R-L load: a) PWM load voltage and current b) Spectra.

VI. Conclusions

Circuit simulation using the evaluation version of PSpice has proven to be an effective tool for learning the behavior of several power electronics circuits. PSpice has been used extensively by students in investigating circuits and in student design projects to determine if a circuit meets design criteria.

References

[1] *PSpice Reference Manual* Version 5.0, MicroSim Corporation, Irvine, CA., 1991.

[2] D. W. Hart, "Using PSpice for undergraduate design projects in power electronics," *ASEE Annual Conf. Proc.*, pp. 1717–1720, June 1991.

[3] V. Agrawal, A. K. Argarwal, and K. Kant, "A study of single-phase to three-phase cycloconverters using PSpice," *IEEE Trans. Industrial Elect.*, vol. 39, no. 2, pp. 141–148, Apr. 1992.

[4] L.J. Giacoletto, "Simple SCR and Triac PSpice computer models," *IEEE Trans. Industrial Elect.*, vol. 36, no. 3, pp. 451–455, Aug. 1989.

[5] M. J. Fisher, *Power Electronics.* – Boston: PWS-Kent, 1991.

[6] N. Mohan, T. M. Undeland, and W. T. Robbins, *Power Electronics: Converters, Applications, and Design.* New York: Wiley, 1989.

[7] M. H. Rashid, *Power Electronics: Circuits, Devices, and Applications.* Englewood Cliffs, NJ: Prentice-Hall, 1988.



Daniel W. Hart (S'69–M'85–SM'92) was born in Woodstock, IL, on January 16, 1948. He received the BSEE degree from Valparaiso University, Valpariso, IN, in 1970, the MSE and Ph.D. degrees from Purdue University, Lafayette, IN, in 1975 and 1985, respectively.

He is currently an Associate Professor of Electrical and Computer Engineering at Valparaiso University, where he has been teaching at the undergraduate level for twelve years. Prior to that, he was a design engineer in the power industry for seven years. His interests include power electronics, power systems, and undergraduate education.

An Early Introduction to Circuit Simulation Techniques

Lawrence T. Pillage, *Member, IEEE*

Abstract—**Introducing simulation tools such as SPICE at the undergraduate level can sometimes cause the students to lack an appreciation for circuit theory and solving circuits. At the University of Texas, some of our undergraduates question the need to solve circuit problems by hand when SPICE can easily do the job for them. In order to motivate circuit theory, and to instill an appreciation for the limitations of powerful computer aids such as SPICE, we are introducing circuit simulation at the freshman level as part of a pilot course, "Introduction to Electrical and Computer Engineering."**

I. Introduction

The capabilities and the availability of circuit simulation software result in most electrical engineering students using a circuit simulator before they have enough background to understand anything about the inner workings of one. This ease of acquiring circuit solutions can cause the students to lose appreciation for their linear circuit theory course(s). Our experience is that some of the students question the assignment of homework problems which require hand-analyses of circuits since SPICE can solve these problems for them. And, keeping circuit simulators a secret is not possible, since even our freshman have heard of SPICE and its powerful capabilities.

As part of a pilot-project course, *Introduction to Electrical and Computer Engineering*, we are introducing the freshmen to SPICE and computer-based methods for solving circuits. This introduction is, of course, somewhat superficial; however, it stresses the importance of knowing the underlying circuit analysis techniques, particularly for understanding the limitations of these and other software tools. This introduction also helps to expose the students to the subject of computer-based problem solving. Namely, it provides practical applications for numerical techniques such as numerical integration and Newton's method which are introduced in their calculus courses.

The following sections sketch out our approach to introducing circuit simulation to the freshman at the University of Texas at Austin. First, we address the power of computer-based techniques for solving dc circuits. Then, we can consider numerical approximations for performing transient analysis of circuits which includes an introduction to circuit models, both linear and nonlinear. These circuit simulation topics are covered as part of the introduction to linear circuit analysis in our freshman course. This material also provides a link to other topics in the course, namely, the concept of delay in digital logic, and numerical methods.

II. DC Nodal Analysis

Nodal analysis is perhaps the most popular method for formulating circuit equations, especially when they are formed and solved using a computer. One potential problem with nodal analysis is the perception that voltage sources are not readily handled. And it is desirable to introduce voltage source models to freshman since, unlike current sources, they can easily be related to a physical device (the battery). For circuit simulation, however, voltage sources actually simplify the complexity of the circuit equations solution by eliminating one node voltage variable. In particular, grounded voltage sources are trivially handled by nodal analysis, and therefore, we consider only grounded voltage sources and batteries for the freshman class. The students, however, seem highly motivated to learn nodal analysis once they realize that it is the method used in SPICE.

Manuscript received June 1992. This work was supported in part by the National Science Foundation under Grant USE-9150505.

The author is with the Department of Electrical and Computer Engineering, the University of Texas at Austin, Austin, TX, 78712.

IEEE Log Number 9202851.